
1. [bookmark: _Toc122434485][bookmark: _Toc89694466]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 11.1.6 which is part of the NR5GC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
Contact:	Pengchong Shen
	shenpengchong@datangmobile.cn
0. [bookmark: _Toc89694467]Verification Test Summary
Test Case: 	11.1.6
ATS Version:	iwd-TTCN3-B2020-09_D21wk37
System Simulator used:	Datang LinkTester 5G Protocol Conformance Test System ECT9610
UE used:	HiSilicon Balong 5000
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc89694468]Corrections required
1. [bookmark: _Toc67572157][bookmark: _Toc89694469][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]Change 1
	Function name
	fl_TC_11_1_6_NR5GC_TestBody

	Reason for change
	1. Acoording to 38.331 clause 5.3.8.3, UE will ignore any field included in RRCRelease message except waitTime if the AS security is not activated, so it is should add security activate process after step 5.

	Summary of change
	1. Add f_NR_RRC_ActivateSecurity  after step5.

	TTCN module
	EPS_Fallback_NR5GC_NR.ttcn

	MCC160 Comment
	



Before Change
	  function fl_TC_11_1_6_NR5GC_TestBody() runs on NR5GC_PTC
  {
    var EUTRA_NR_CoOrd_SysInfo_Type v_CoOrd_SysInfo := f_EUTRA_NR_WaitForCoOrd_SysInfo(EUTRA);
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var NG_NAS_GutiParameters_Type v_GutiParams := f_NR5GC_CellInfo_GetGutiParameters (nr_Cell1);
    var template (present) NG_MobileIdentity v_S_TMSI := f_NG_S_TMSI2MobileIdentity(omit, v_GutiParams);
    var template (value) RedirectedCarrierInfo v_RedirectedCarrierInfo;

    //@siclog "Step 2" siclog@
    // The SS transmits a Paging message on NR Cell 1
    f_NR_Paging_Def(nr_Cell1, RRC_IDLE);

    //@siclog "Steps 3-4" siclog@
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell1, tsc_NR_RRC_TI_Def, ?);
    
    //@siclog "Step 5" siclog@
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu,
                                      cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext),
                                      v_S_TMSI,
                                      ?,
                                      -, ?)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    
    // @siclog "Step 5A" siclog@
    f_NR_SetCellPower (nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2); //@sic R5-210133 sic@
    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger("Change Power Level"));  // @sic R5-216190 sic@

    //@siclog "Step 6" siclog@
    //The SS transmits an RRCRelease message (IE redirectedCarrierInfo including EUTRA cell).
    v_RedirectedCarrierInfo := cs_NR_RedirectedCarrierInfoToEUTRA(v_CoOrd_SysInfo.Eutra[0].Arfcn);
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, -, cs_38508_RRCRelease(tsc_NR_RRC_TI_Def, -,
                                                                     v_RedirectedCarrierInfo))); // @sic R5s210662 sic@
    f_NR_RRCRelease_Local(nr_Cell1, f_NR_GetNextSendOccasion_DL(nr_Cell1)); // @sic R5s210662 sic@

    // Send authentication paramters to EUTRA
    f_NR5GC_Send_GutiParametersToEUTRA(nr_Cell1);
    
    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA, "Postamble");
  }



After Change
	   function fl_TC_11_1_6_NR5GC_TestBody() runs on NR5GC_PTC
  {
    var EUTRA_NR_CoOrd_SysInfo_Type v_CoOrd_SysInfo := f_EUTRA_NR_WaitForCoOrd_SysInfo(EUTRA);
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var NG_NAS_GutiParameters_Type v_GutiParams := f_NR5GC_CellInfo_GetGutiParameters (nr_Cell1);
    var template (present) NG_MobileIdentity v_S_TMSI := f_NG_S_TMSI2MobileIdentity(omit, v_GutiParams);
    var template (value) RedirectedCarrierInfo v_RedirectedCarrierInfo;

    //@siclog "Step 2" siclog@
    // The SS transmits a Paging message on NR Cell 1
    f_NR_Paging_Def(nr_Cell1, RRC_IDLE);

    //@siclog "Steps 3-4" siclog@
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell1, tsc_NR_RRC_TI_Def, ?);
    
    //@siclog "Step 5" siclog@
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu,
                                      cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext),
                                      v_S_TMSI,
                                      ?,
                                      -, ?)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    f_NR_RRC_ActivateSecurity (nr_Cell1, v_ReceivedMsg.SecurityProtection.NasCount);
    // @siclog "Step 5A" siclog@
    f_NR_SetCellPower (nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2); //@sic R5-210133 sic@
    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger("Change Power Level"));  // @sic R5-216190 sic@

    //@siclog "Step 6" siclog@
    //The SS transmits an RRCRelease message (IE redirectedCarrierInfo including EUTRA cell).
    v_RedirectedCarrierInfo := cs_NR_RedirectedCarrierInfoToEUTRA(v_CoOrd_SysInfo.Eutra[0].Arfcn);
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, -, cs_38508_RRCRelease(tsc_NR_RRC_TI_Def, -,
                                                                     v_RedirectedCarrierInfo))); // @sic R5s210662 sic@
    f_NR_RRCRelease_Local(nr_Cell1, f_NR_GetNextSendOccasion_DL(nr_Cell1)); // @sic R5s210662 sic@

    // Send authentication paramters to EUTRA
    f_NR5GC_Send_GutiParametersToEUTRA(nr_Cell1);
    
    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA, "Postamble");
  }


1. [bookmark: _Toc67572159][bookmark: _Toc89694470]Change 2
	Function name
	fl_TC_11_1_6_NR5GC_EUTRA_TestBody

	Reason for change
	1. According to 36.508 Table 4.5A.7.3-1, there is no other reconfiguration process between step 7 and step 13.
2. Acoording to TS 23.502 clause 4.1.1.2.2  step13. It should to add PDN CONNECTIVITY REQUEST process.

	Summary of change
	1. Delete f_EUTRA38_MultiPDN_Activate_SRB2_DRB_SendRrcMsg after step 21
2. Add PDN CONNECTIVITY REQUEST process after step 20.

	TTCN module
	EPS_Fallback_NR5GC_EUTRA.ttcn

	MCC160 Comment
	



Before Change
	  function fl_TC_11_1_6_NR5GC_EUTRA_TestBody() runs on EUTRA_Multi5GS_PTC
  {
    var NasCountInfo_Type v_NasCountInfo;
    
    // Send EUTRA frequency to NR and wait for NR info
    f_EUTRA_NR_SendSysinfoCoOrd(NR, f_EUTRA38_IRAT_ListOfFrequencies({eutra_Cell1}));
    
    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Change Power Level");  // @sic R5-216190 sic@
    // @siclog "Step 5A" siclog@
    f_EUTRA38_SetCellPower(eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE);

    fl_TAUReject_AttachWithoutN26 ();
    // Wait here for optional re-register
    f_EUTRA_DelayForUserPlaneSignalling (waitForIMS);
    // then tell IMS to continue with MT call
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]);

    // @siclog "Step 21" siclog@
    // Step 7 of 4.5A.6.3-1
    f_EUTRA38_MultiPDN_RRCConnectionReconfiguration_Common(eutra_Cell1,
                                                           tsc_RRC_TI_Def,
                                                           {cs_508_SRB2_ToAddMod_DEFAULT},
                                                           {cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB1)});
    
    // Get the latest NAS count FFS
    v_NasCountInfo := f_EUTRA_NB_SS_NAS_CountGet(NASCTRL);

    // @siclog "Step 22" siclog@
    f_EUTRA38_MultiPDN_Activate_SRB2_DRB_SendRrcMsg(eutra_Cell1,
                                                  tsc_RRC_TI_Def,
                                                  v_NasCountInfo.UL);
    
    // @siclog "Step 22" siclog@
    f_EUTRA_Activate_SRB2_DRB_ReceiveRrcAndNasMsg(eutra_Cell1,
                                                  tsc_RRC_TI_Def);

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);

    // @siclog "Step 26 - 28" siclog@
    f_EUTRA38_MultiPDN_ActivateDedicatedEpsBearer_Speech(eutra_Cell1);
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]);
    
    // Test body finishes on IMS PTC
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("Postamble"));
  }



After Change
	   function fl_TC_11_1_6_NR5GC_EUTRA_TestBody() runs on EUTRA_Multi5GS_PTC
  {
var NasCountInfo_Type v_NasCountInfo;
    var EPS_BearerIdentity v_EpsBearerId;
    var SRB_COMMON_IND v_ReceivedAsp;
timer t_wait := 5.0;
    
    // Send EUTRA frequency to NR and wait for NR info
    f_EUTRA_NR_SendSysinfoCoOrd(NR, f_EUTRA38_IRAT_ListOfFrequencies({eutra_Cell1}));
    
    f_EUTRA_NR_WaitForCoOrd_Trigger (NR, "Change Power Level");  // @sic R5-216190 sic@
    // @siclog "Step 5A" siclog@
    f_EUTRA38_SetCellPower(eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE);

fl_TAUReject_AttachWithoutN26 ();
alt {
      [] SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,
                                           cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered, cr_PDN_CONNECTIVITY_REQUEST_N1ToS1(tsc_PdnRequest_Handover, ?)))) -> value v_ReceivedAsp
        {
          v_EpsBearerId := f_EUTRA38_MultiPDN_EstablishAdditionalPDN_Step2_FB(eutra_Cell1, v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.pDN_CONNECTIVITY_REQUEST);          a_EUTRA_IdleUpdated_ConnectToAdditionalPDN_Step3(eutra_Cell1);
          a_EUTRA38_MultiPDN_ConnectToAdditionalPDN_Step4(eutra_Cell1, v_EpsBearerId);
        }
      [] t_wait.timeout
        {
        }
     }
    // Wait here for optional re-register
    f_EUTRA_DelayForUserPlaneSignalling (waitForIMS);
    // then tell IMS to continue with MT call
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]);

    // @siclog "Step 21" siclog@
    // Step 7 of 4.5A.6.3-1
    f_EUTRA38_MultiPDN_RRCConnectionReconfiguration_Common(eutra_Cell1,
                                                           tsc_RRC_TI_Def,
                                                           {cs_508_SRB2_ToAddMod_DEFAULT},
                                                           {cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB1)});
        
    // @siclog "Step 22" siclog@
    f_EUTRA_Activate_SRB2_DRB_ReceiveRrcAndNasMsg(eutra_Cell1,
                                                  tsc_RRC_TI_Def);

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);

    // @siclog "Step 26 - 28" siclog@
    f_EUTRA38_MultiPDN_ActivateDedicatedEpsBearer_Speech(eutra_Cell1);
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]);
    
    // Test body finishes on IMS PTC
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger("Postamble"));
  }


1. [bookmark: _Toc89694471]Change 3
	Function name
	f_TC_11_1_6_NR5GC_IMS1

	Reason for change
	1. v_SipUrl can not be initialized until fl_IMSFallback_PreambleRegistrations is run.

	Summary of change
	1. Add f_IMS_IPCAN_SendCoOrdMsg after step 11.

	TTCN module
	EPS_Fallback_NR5GC_IMS.ttcn

	MCC160 Comment
	



Before Change
	[bookmark: OLE_LINK5][bookmark: OLE_LINK6]  function f_TC_11_1_6_NR5GC_IMS1()runs on IMS_PTC
  {
    var IMS_IPCAN_Coordination_MSG v_IMS_IPCAN_CoordMsg;
    var INVITE_Request v_InviteRequest;
    var template (present) SipUrl v_SipUrl := cr_SipUri_HostPort(f_IMS_PTC_ImsInfo_GetHomeDomainName());
    var IMS_DATA_REQ v_IMS_REGISTER_REQ_ByRef;
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdpVal; /* @sic R5w140112: v_InviteRequestWithSdpVal instead of v_InviteRequest sic@ */
    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;              /* @sic R5w140112 sic@ */
    var IMS_Response_Type v_SessionInProgress183;
    var SipUrl v_ContactSipUrlUE;
    
    fl_IMSFallback_PreambleRegistrations();

    // Wait for indication RAT has changed to EUTRA
    OtherIPCAN.receive(cmr_IPCAN_IMS_Response) -> value v_IMS_IPCAN_CoordMsg;
    f_IMS_PTC_SetRanType(v_IMS_IPCAN_CoordMsg.IPCAN_IMS_Response.IpcanInfo.RanType);
    
    // Exception: Optional C.46 before Invite Request
    alt {
      [] a_IMS_ReceiveRequest(car_IMS_Register_Request(cr_REGISTER_Request(v_SipUrl)), v_IMS_REGISTER_REQ_ByRef) {

         IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_REGISTER_REQ_ByRef.RoutingInfo),
                                     cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_IMS_REGISTER_REQ_ByRef.Request.Register))));
         repeat;
        
        }
      [] OtherIPCAN.receive(cmr_IMS_IPCAN_Trigger) {}
    }
        
    // @siclog "Step 21 - 28" siclog@
    v_InviteRequestWithSdpVal := f_IMS_MTCallSetup_BuildInviteRequestWithSdp();
    v_InviteRequest := v_InviteRequestWithSdpVal.Invite;
    v_ResponseWithSdp := f_IMS_MTCallSetup_AnnexC11_Steps1_4(v_InviteRequestWithSdpVal);
    v_SessionInProgress183 := v_ResponseWithSdp.Response;
    v_ContactSipUrlUE := f_MessageHeader_GetContactSipUrl(v_SessionInProgress183.msgHeader);
    
    // Step 5 - 6. Send PRACK, Receive 200 OK
    f_IMS_MTCallSetup_SendPRACK_ReceiveOK(v_InviteRequest, v_SessionInProgress183.msgHeader);     /* @sic R5-181141: steps 5-6 before sending the trigger sic@ */

    // Inform EUTRA: IMS Call established.
    f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);
    
    //Wait for Co-ordination message from EUTRA
    f_IMS_IPCAN_WaitForTrigger(OtherIPCAN);

    // Step 7 - 13
    f_IMS_MTCallSetup_AnnexC11_Steps7_13(v_InviteRequestWithSdpVal, v_ResponseWithSdp);

    // @siclog "Step 41" siclog@
    f_Delay(1.0);
    
    // @siclog "Step 42" siclog@
    f_IMS_CallReleaseMT(v_ContactSipUrlUE);
    f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);

    f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN(); // @sic R5s210662 sic@
  }



After Change
	   function f_TC_11_1_6_NR5GC_IMS1()runs on IMS_PTC
  {
    var IMS_IPCAN_Coordination_MSG v_IMS_IPCAN_CoordMsg;
    var INVITE_Request v_InviteRequest;
    var template (present) SipUrl v_SipUrl;
    var IMS_DATA_REQ v_IMS_REGISTER_REQ_ByRef;
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdpVal; /* @sic R5w140112: v_InviteRequestWithSdpVal instead of v_InviteRequest sic@ */
    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;              /* @sic R5w140112 sic@ */
    var IMS_Response_Type v_SessionInProgress183;
    var SipUrl v_ContactSipUrlUE;
    
    fl_IMSFallback_PreambleRegistrations();
v_SipUrl := cr_SipUri_HostPort(f_IMS_PTC_ImsInfo_GetHomeDomainName());
    // Wait for indication RAT has changed to EUTRA
    OtherIPCAN.receive(cmr_IPCAN_IMS_Response) -> value v_IMS_IPCAN_CoordMsg;
    f_IMS_PTC_SetRanType(v_IMS_IPCAN_CoordMsg.IPCAN_IMS_Response.IpcanInfo.RanType);
    
    // Exception: Optional C.46 before Invite Request
    alt {
      [] a_IMS_ReceiveRequest(car_IMS_Register_Request(cr_REGISTER_Request(v_SipUrl)), v_IMS_REGISTER_REQ_ByRef) {

         IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_REGISTER_REQ_ByRef.RoutingInfo),
                                     cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_IMS_REGISTER_REQ_ByRef.Request.Register))));
         repeat;
        
        }
      [] OtherIPCAN.receive(cmr_IMS_IPCAN_Trigger) {}
    }
        
    // @siclog "Step 21 - 28" siclog@
    v_InviteRequestWithSdpVal := f_IMS_MTCallSetup_BuildInviteRequestWithSdp();
    v_InviteRequest := v_InviteRequestWithSdpVal.Invite;
    v_ResponseWithSdp := f_IMS_MTCallSetup_AnnexC11_Steps1_4(v_InviteRequestWithSdpVal);
    v_SessionInProgress183 := v_ResponseWithSdp.Response;
    v_ContactSipUrlUE := f_MessageHeader_GetContactSipUrl(v_SessionInProgress183.msgHeader);
    
    // Step 5 - 6. Send PRACK, Receive 200 OK
    f_IMS_MTCallSetup_SendPRACK_ReceiveOK(v_InviteRequest, v_SessionInProgress183.msgHeader);     /* @sic R5-181141: steps 5-6 before sending the trigger sic@ */

    // Inform EUTRA: IMS Call established.
    f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);
    
    //Wait for Co-ordination message from EUTRA
    f_IMS_IPCAN_WaitForTrigger(OtherIPCAN);

    // Step 7 - 13
    f_IMS_MTCallSetup_AnnexC11_Steps7_13(v_InviteRequestWithSdpVal, v_ResponseWithSdp);

    // @siclog "Step 41" siclog@
    f_Delay(1.0);
    
    // @siclog "Step 42" siclog@
    f_IMS_CallReleaseMT(v_ContactSipUrlUE);
    f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);

    f_IMS_IpcanReleaseWithOptionalDeregistrationOtherIPCAN(); // @sic R5s210662 sic@
  }


1. [bookmark: _Toc89694472]Change 4
	Function name
	fl_TAUReject_AttachWithoutN26

	Reason for change
	1. According to 24.301, cl. 5.5.1.2.2, the UE shall include the valid GUTI into the EPS mobile identity IE, not mapped NG_GUTI.

	Summary of change
	1. Change the forth parameter of cdr_ATTACH_REQUEST_N1ToS1 to ‘v_GutiParams’.

	TTCN module
	EPS_Fallback_NR5GC_EUTRA.ttcn

	MCC160 Comment
	



Before Change
	  function fl_TAUReject_AttachWithoutN26 () runs on EUTRA_Multi5GS_PTC
  {
    var SRB_COMMON_IND v_ReceivedAsp;
    var NAS_MSG_Indication_Type v_NasInd;
    var template (present) NAS_MSG_Indication_Type v_NAS_Expected;
    var EUTRA_NR_CoOrd_GUTI_Type v_CoOrd_Guti;
    var template (value) MobileIdentity v_MappedGUTI;
    var GutiParameters_Type v_GutiParams;
    var template AdditionalUpdateType v_AdditionalUpdateType;
    var NAS_PlmnId v_PLMN;
    var TrackingAreaCode v_Tac;
    var template TrackingAreaId v_TAI:= omit;
    var template (present) NAS_KsiValue v_KsiValue;
    var EUTRA_NR_PduSessionInfoList_Type v_PduSessionInfoList;
    var template (omit) integer v_SessionId;
        
    v_CoOrd_Guti := f_EUTRA_NR_WaitForCoOrd_GutiParams(NR);
    v_MappedGUTI := f_EUTRA38_NR_Convert5G_CoOrdToEPS_GUTI (v_CoOrd_Guti, omit);
    v_PduSessionInfoList := f_EUTRA38_WaitForPDUSessionInfoList(eutra_Cell1); // @sic R5s210149 sic@
    
    // Tell IMS PTC the RAT has changed
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1], cms_IPCAN_IMS_QueryResponse(f_EUTRA_IMS_IpcanInfo()));

    v_AdditionalUpdateType := f_GetAdditionalUpdateType (NORMAL);
    v_KsiValue := f_EUTRA_SecurityKSIasme_Get();
    v_GutiParams:= f_EUTRA_CellInfo_GetGuti(eutra_Cell1);
    v_PLMN := f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity);
    v_Tac := f_EUTRA_CellInfo_GetTAC(eutra_Cell1);
    v_TAI := cr_TAI(v_PLMN, bit2oct(v_Tac));
   
    // @siclog "Step 7 - 8" siclog@
    v_NAS_Expected := (cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                          cr_TAU_Request_N1ToS1(f_GetEPSTAUType(NORMAL),
                                                                v_KsiValue,
                                                                cr_DRXparameter_Any ('5C'O) ifpresent,
                                                                v_AdditionalUpdateType,
                                                                v_MappedGUTI,
                                                                v_TAI,
                                                                tsc_EpsUpdate_Active,
                                                                -, -,
                                                                omit,
                                                                cr_UeRadioCap ifpresent)),
                       cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,
                                                         cdr_ATTACH_REQUEST_N1ToS1(f_GetEAttachType(NORMAL),
                                                                                   v_AdditionalUpdateType,
                                                                                   v_KsiValue,
                                                                                   v_MappedGUTI,
                                                                                   v_TAI),
                                                         cr_PDN_CONNECTIVITY_REQUEST_N1ToS1(tsc_PdnRequest_Handover,
                                                                                            f_GetPdnType())));
    
    // @siclog "Step 8a1/8b1" siclog@
    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_Cell1,
                                               tsc_RRC_TI_Def,
                                               ?,
                                               v_NAS_Expected);

    if (ischosen (v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST)) {
        // @siclog "Step 8b2" siclog@
        SRB.send(cas_SRB1_NasPdu_REQ(eutra_Cell1,
                                   cs_TimingInfo_Now,
                                   cs_NAS_Request(tsc_SHT_NoSecurityProtection,
                                                  cs_508_TRACKING_AREA_UPDATE_REJECT(tsc_EmmCause_UEIdentityNotDerived))));

        // @siclog "Step 8b3" siclog@
        SRB.receive(car_SRB1_NasPdu_IND(eutra_Cell1,
                                           cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),
                                                                                   cdr_ATTACH_REQUEST_IMSIonly(f_GetEAttachType(NORMAL),
                                                                                   v_AdditionalUpdateType,
                                                                                   v_TAI),
                                                         cr_PDN_CONNECTIVITY_REQUEST_N1ToS1(tsc_PdnRequest_Handover,
                                                                                            f_GetPdnType()))))
                            -> value v_ReceivedAsp;
        v_NasInd := v_ReceivedAsp.Signalling.Nas[0];
    }

    // @siclog "Step 9 - 20" siclog@
    v_SessionId := f_ProtocolConfigOptions_GetSessionId(v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions);
    if (not isvalue (v_SessionId)) {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Id not present in PCO");
    }

    f_EUTRA38_MultiPDN_InitialRegistration_Step5_16 (eutra_Cell1, TESTMode_OFF, v_NasInd, -, v_PduSessionInfoList); // @sic R5s210149 sic@

  }



After Change
	   function fl_TAUReject_AttachWithoutN26 () runs on EUTRA_Multi5GS_PTC
  {
    var SRB_COMMON_IND v_ReceivedAsp;
    var NAS_MSG_Indication_Type v_NasInd;
    var template (present) NAS_MSG_Indication_Type v_NAS_Expected;
    var EUTRA_NR_CoOrd_GUTI_Type v_CoOrd_Guti;
    var template (value) MobileIdentity v_MappedGUTI;
    var GutiParameters_Type v_GutiParams;
    var template AdditionalUpdateType v_AdditionalUpdateType;
    var NAS_PlmnId v_PLMN;
    var TrackingAreaCode v_Tac;
    var template TrackingAreaId v_TAI:= omit;
    var template (present) NAS_KsiValue v_KsiValue;
    var EUTRA_NR_PduSessionInfoList_Type v_PduSessionInfoList;
    var template (omit) integer v_SessionId;
        
    v_CoOrd_Guti := f_EUTRA_NR_WaitForCoOrd_GutiParams(NR);
    v_MappedGUTI := f_EUTRA38_NR_Convert5G_CoOrdToEPS_GUTI (v_CoOrd_Guti, omit);
    v_PduSessionInfoList := f_EUTRA38_WaitForPDUSessionInfoList(eutra_Cell1); // @sic R5s210149 sic@
    
    // Tell IMS PTC the RAT has changed
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1], cms_IPCAN_IMS_QueryResponse(f_EUTRA_IMS_IpcanInfo()));

    v_AdditionalUpdateType := f_GetAdditionalUpdateType (NORMAL);
    v_KsiValue := f_EUTRA_SecurityKSIasme_Get();
    v_GutiParams:= f_EUTRA_CellInfo_GetGuti(eutra_Cell1);
    v_PLMN := f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity);
    v_Tac := f_EUTRA_CellInfo_GetTAC(eutra_Cell1);
    v_TAI := cr_TAI(v_PLMN, bit2oct(v_Tac));
   
    // @siclog "Step 7 - 8" siclog@
    v_NAS_Expected := (cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                          cr_TAU_Request_N1ToS1(f_GetEPSTAUType(NORMAL),
                                                                v_KsiValue,
                                                                cr_DRXparameter_Any ('5C'O) ifpresent,
                                                                v_AdditionalUpdateType,
                                                                v_MappedGUTI,
                                                                v_TAI,
                                                                tsc_EpsUpdate_Active,
                                                                -, -,
                                                                omit,
                                                                cr_UeRadioCap ifpresent)),
                       cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,
                                                         cdr_ATTACH_REQUEST_N1ToS1(f_GetEAttachType(NORMAL),
                                                                                   v_AdditionalUpdateType,
                                                                                   v_KsiValue,
                                                                                   f_GutiParameters2MobileIdentity(omit, v_GutiParams),
                                                                                   v_TAI),
                                                         cr_PDN_CONNECTIVITY_REQUEST_N1ToS1(tsc_PdnRequest_Handover,
                                                                                            f_GetPdnType())));
    
    // @siclog "Step 8a1/8b1" siclog@
    v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_Cell1,
                                               tsc_RRC_TI_Def,
                                               ?,
                                               v_NAS_Expected);

    if (ischosen (v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST)) {
        // @siclog "Step 8b2" siclog@
        SRB.send(cas_SRB1_NasPdu_REQ(eutra_Cell1,
                                   cs_TimingInfo_Now,
                                   cs_NAS_Request(tsc_SHT_NoSecurityProtection,
                                                  cs_508_TRACKING_AREA_UPDATE_REJECT(tsc_EmmCause_UEIdentityNotDerived))));

        // @siclog "Step 8b3" siclog@
        SRB.receive(car_SRB1_NasPdu_IND(eutra_Cell1,
                                           cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),
                                                                                   cdr_ATTACH_REQUEST_IMSIonly(f_GetEAttachType(NORMAL),
                                                                                   v_AdditionalUpdateType,
                                                                                   v_TAI),
                                                         cr_PDN_CONNECTIVITY_REQUEST_N1ToS1(tsc_PdnRequest_Handover,
                                                                                            f_GetPdnType()))))
                            -> value v_ReceivedAsp;
        v_NasInd := v_ReceivedAsp.Signalling.Nas[0];
    }

    // @siclog "Step 9 - 20" siclog@
    v_SessionId := f_ProtocolConfigOptions_GetSessionId(v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions);
    if (not isvalue (v_SessionId)) {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Id not present in PCO");
    }

    f_EUTRA38_MultiPDN_InitialRegistration_Step5_16 (eutra_Cell1, TESTMode_OFF, v_NasInd, -, v_PduSessionInfoList); // @sic R5s210149 sic@

  }


1. [bookmark: _Toc89694473]Change 5
	Function name
	cdr_ATTACH_REQUEST_N1ToS1

	Reason for change
	1. According to 24.301, cl. 5.5.1.2.2, the UE shall include the valid GUTI into the EPS mobile identity IE.

	Summary of change
	1. Instead additionalGuti to epsMobileIdentity.

	TTCN module
	EPS_Fallback_NR5GC_EUTRA.ttcn

	MCC160 Comment
	



Before Change
	  template (present) NAS_UL_Message_Type cdr_ATTACH_REQUEST_N1ToS1(template (present) NAS_AttDetValue_Type p_AttachTypeValue,
                                                                   template AdditionalUpdateType p_AdditionalUpdateType,
                                                                   template (present) NAS_KsiValue p_KsiValue,
                                                                   template (present) MobileIdentity p_Guti,
                                                                   template TrackingAreaId p_TAI)
    modifies cr_508_ATTACH_REQUEST :=
  { // FFS: OldGUTI is not included in this message definition!
    aTTACH_REQUEST := {
      nasKeySetId              := cr_NAS_KeySetIdentifier(p_KsiValue, tsc_Spare1),
      additionalGuti           := p_Guti,
      lastVisitedRegisteredTai := p_TAI,
      oldGUTI_Type               := cr_GUTI_Type_Native
    }
  }



After Change
	   template (present) NAS_UL_Message_Type cdr_ATTACH_REQUEST_N1ToS1(template (present) NAS_AttDetValue_Type p_AttachTypeValue,
                                                                   template AdditionalUpdateType p_AdditionalUpdateType,
                                                                   template (present) NAS_KsiValue p_KsiValue,
                                                                   template MobileIdentity p_Guti,
                                                                   template TrackingAreaId p_TAI)
    modifies cr_508_ATTACH_REQUEST :=
  { // FFS: OldGUTI is not included in this message definition!
    aTTACH_REQUEST := {
      nasKeySetId              := cr_NAS_KeySetIdentifier(p_KsiValue, tsc_Spare1),
      epsMobileIdentity           := p_Guti,
      lastVisitedRegisteredTai := p_TAI,
      oldGUTI_Type               := cr_GUTI_Type_Native
    }
  }


1. [bookmark: _Toc89694474]Change 6
	Function name
	cr_PDN_CONNECTIVITY_REQUEST_N1ToS1

	Reason for change
	1. According to 38.523-1 table 11.1.5 3.3-7 and 36.508 table 4.7.2-20. ESM information transfer flag should be set to ‘Not present or any allowed value’.

	Summary of change
	1. Change the fifth parameter of cr_PDN_CONNECTIVITY_REQUEST_Common to ‘*’

	TTCN module
	EPS_Fallback_NR5GC_EUTRA.ttcn

	MCC160 Comment
	



Before Change
	  template (present) NAS_UL_Message_Type cr_PDN_CONNECTIVITY_REQUEST_N1ToS1(template (present) B3_Type p_RequestTypeValue,
                                                                            template (present) PdnTypeValue p_PDNTypeValue,
                                                                            template ProtocolConfigOptions p_PCO := cr_ProtocolConfigOptionsAny ifpresent,
                                                                            template ExtdProtocolConfigOptions p_ExtdProtocolConfigOptions := cr_ExtdPCOAny ifpresent) :=
    cr_PDN_CONNECTIVITY_REQUEST_Common ('0'H,
                                        ?,
                                        p_PDNTypeValue,
                                        p_RequestTypeValue,
                                        -, -,
                                        p_PCO,
                                        -, -, -,
                                        p_ExtdProtocolConfigOptions);



After Change
	   template (present) NAS_UL_Message_Type cr_PDN_CONNECTIVITY_REQUEST_N1ToS1(template (present) B3_Type p_RequestTypeValue,
                                                                            template (present) PdnTypeValue p_PDNTypeValue,
                                                                            template ProtocolConfigOptions p_PCO := cr_ProtocolConfigOptionsAny ifpresent,
                                                                            template ExtdProtocolConfigOptions p_ExtdProtocolConfigOptions := cr_ExtdPCOAny ifpresent) :=
    cr_PDN_CONNECTIVITY_REQUEST_Common ('0'H,
                                        ?,
                                        p_PDNTypeValue,
                                        p_RequestTypeValue,
                                        *, -,
                                        p_PCO,
                                        -, -, -,
                                        p_ExtdProtocolConfigOptions);


1. [bookmark: _Toc89693772][bookmark: _Toc89694475]Change 7
	Function name
	f_EUTRA38_MultiPDN_EstablishAdditionalPDN_Step2_FB

	Reason for change
	1. In EPS Fall Back, The internet type IP link already exists, it should use the f_IP_ChangeDrbMappingEUTRA function to establish the IP link.

	Summary of change
	1. Add new function f_EUTRA38_MultiPDN_EstablishAdditionalPDN_Step2_FB

	TTCN module
	EUTRA38_NR5GC_InitialRegistration.ttcn

	MCC160 Comment
	



New function
	[bookmark: OLE_LINK11][bookmark: OLE_LINK12]  function f_EUTRA38_MultiPDN_EstablishAdditionalPDN_Step2_FB(EUTRA_CellId_Type p_CellId,
                                                           PDN_CONNECTIVITY_REQUEST p_PDNConnReq,
                                                           boolean p_ConfigureSRB2 := false,
                                                           template (omit) NasCount_Type p_NasCountUL := omit,
                                                           template (omit) EUTRA_NR_PduSessionInfoList_Type p_PduInfoFromNR := omit) runs on EUTRA_Multi5GS_PTC return EPS_BearerIdentity
  {
    var ProcedureTransactionIdentifier v_PTId_UE;
    var template (omit) ProtocolConfigOptions v_PcoFromUE;
    var template (value) ProtocolConfigOptions v_PcoToUE;
    var template (omit) ExtdProtocolConfigOptions v_ExtdPcoFromUE := omit;
    var template (value) AccessPointName v_APN_Tx := f_EUTRA38_MultiPDN_GetAPN(p_CellId, p_PDNConnReq.accessPointName);
    var PDU_PDN_DNN_Type v_PDN_Type;
    var PDN_Index_Type v_PdnIndex;
    var boolean v_IsIMS := false;
    var EPS_BearerIdentity v_EpsDefaultBearerId;
    var PDN_Address v_PDN_Address;
    var BEARER_CONTEXT_TYPE v_BearerContextNumber;
    var template (omit) ESM_Cause v_ESM_Cause := omit;        // @sic R5s170882 sic@
    var PDN_TypeToBeUsed_Type v_PDN_TypeToBeUsed := f_EUTRA_MobileInfo_GetPdnTypeToBeUsed();   /* @sic R5s170882 sic@ */
    var template (omit) SRB_ToAddModList v_SRB_ToAddModList := omit;
    var template (value) EUTRA38_DRBInfoList_Type v_NewDRBs := {};
    var template (value) RadioBearerList_Type v_SS_ConfigList;
    var template (value) DRB_ToAddModList v_DRB_ToAddModList;
    var EUTRA38_DRBInfoList_Type v_ExistingDRBs := f_EUTRA38_MobileInfo_GetDRBInfoList();
    var integer v_NewDRBId;
    var EUTRA_SecurityParams_Type v_AuthParams := f_EUTRA_Security_Get();
    var boolean v_StartPDN := true; // @sic R5s210149 sic@
    var integer i;

    v_PTId_UE := p_PDNConnReq.procedureTransactionIdentifier;
    v_PDN_Type := f_EUTRA38_MultiPDN_GetPDNType(p_PDNConnReq.accessPointName);   /* get APN without any ESM information transfer @sic R5-206296 sic@ */
    v_PcoFromUE := p_PDNConnReq.protocolConfigurationOptions;
    v_ExtdPcoFromUE := p_PDNConnReq.extdProtocolConfigurationOptions;
    v_PdnIndex := f_GetMultiPdnIndex (v_PDN_Type); // @sic R5-206296 sic@
    v_PDN_Address := f_EUTRA_GetPDNAddress(f_CheckPCOforIPallocationViaNas(v_PcoFromUE), v_PdnIndex);

    select(v_PDN_Type) { // @sic R5-206296 sic@
      case (Internet_DNN) {
        v_BearerContextNumber := DEF_1;
      }
      case (IMS_DNN, Emergency_PDN) { // @sic R5s210149 sic@
        v_IsIMS := true;
        v_BearerContextNumber := DEF_2;
      }
    }

    // get new default DRB and add to existing list   @sic R5-206289 sic@
    v_NewDRBId := f_EUTRA38_GetNextDRBId (v_PDN_Type, true);
    v_EpsDefaultBearerId := int2hex(v_NewDRBId + 4, 1);
    v_NewDRBs[0] := cs_DRB_Info (v_NewDRBId, hex2int(v_EpsDefaultBearerId), true, cs_OneDRB_ConfigAM(v_NewDRBId), cs_508_DRB_ToAddMod_DEFAULT_AM(v_NewDRBId), v_PDN_Type);
    v_ExistingDRBs := f_EUTRA38_DRBInfo_AddListOfEntries (valueof(v_ExistingDRBs), valueof(v_NewDRBs));
    f_EUTRA38_MobileInfo_SetDRBInfoList (valueof(v_ExistingDRBs));

    if (p_ConfigureSRB2) {  // @sic R5-206289 sic@
      v_SRB_ToAddModList := {cs_508_SRB2_ToAddMod_DEFAULT};
      // Need to configure the whole list
      for (i:=0;  i < lengthof(v_ExistingDRBs); i :=i+1) {
        v_SS_ConfigList[i] := v_ExistingDRBs[i].SS_Config;
        v_DRB_ToAddModList[i] := v_ExistingDRBs[i].DRB_ToAdd;
      }
      // If called after TAU, then NAS Count MUST be a value
      // This will also configure all DRBs
      v_AuthParams := f_EUTRA_RRC_ActivateSecurity(p_CellId, v_AuthParams, valueof(p_NasCountUL), v_SS_ConfigList);
      f_EUTRA_Security_Set(v_AuthParams);
    } else {
      // Only need to configure the latest DRB (all others already configured)
      v_SS_ConfigList[0] := v_NewDRBs[0].SS_Config;
      v_DRB_ToAddModList[0] := v_NewDRBs[0].DRB_ToAdd;
      f_EUTRA_SS_CommonRadioBearerConfig(p_CellId, valueof(v_SS_ConfigList));
      f_EUTRA_SS_RRC_RestartSecurity(p_CellId);
    }
    
    if (isvalue(p_PduInfoFromNR)) { // @sic R5s210149 sic@
    // The UE has just moved from NR. If so, data from NR should be used, rather than created anew
      for (i := 0; i < lengthof (p_PduInfoFromNR); i := i + 1) {
        if (v_PdnIndex == valueof(p_PduInfoFromNR[i].PdnIndex)) {
          // Data from NR should be used, rather than created anew
          v_StartPDN := false; // no need to change DRB mapping as this has already been done in calling function
          break;
        }
      }
    }
    if (v_StartPDN) { // completely new PDN connection @sic R5s210149 sic@
      f_IP_ChangeDrbMappingEUTRA(IP, p_CellId, v_NewDRBId, v_PdnIndex);
    }
    
    v_PcoToUE := f_NAS_ProtocolConfigOptionsTX(f_EUTRA38_MultiPDN_GetPCOs(p_CellId,
                                                                          v_PdnIndex,
                                                                          v_IsIMS,
                                                                          v_EpsDefaultBearerId,
                                                                          v_ExtdPcoFromUE,
                                                                          v_PcoFromUE));

    if (isvalue (v_ExtdPcoFromUE)) {
      // FFS - At the moment, 36.508 table 4.7.3-6 specifies ExtdPCO should be not present
    }

    if (pc_IPv6 and pc_IPv4) {              /* @sic R5s170882 sic@ */
      select (v_PDN_TypeToBeUsed) {
        case (forceIPv4only)          { v_ESM_Cause := cs_ESM_Cause_tv('00110010'B); }     // PDN type IPv4 only allowed
        case (forceIPv6only)          { v_ESM_Cause := cs_ESM_Cause_tv('00110011'B); }     // PDN type IPv6 only allowed
        case (pdnTypeAsSupportedByUE) { v_ESM_Cause := omit; }
      }
    }

    /* Step 8 of procedure in 36.508 cl. 4.5.3.3 */
    f_EUTRA38_MultiPDN_RRCConnectionReconfiguration_Common(p_CellId,
                                                           tsc_RRC_TI_Def,
                                                           v_SRB_ToAddModList,
                                                           v_DRB_ToAddModList,
                                                           1,
                                                           0,
        {cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_508_ActivateDefEpsBearerContextRequest(v_EpsDefaultBearerId,
                                                                                                       v_PTId_UE,
                                                                                                       v_APN_Tx,    // @sic R5-206296 sic@
                                                                                                       v_PDN_Address,
                                                                                                       v_ESM_Cause, /* @sic R5s170882 sic@ */
                                                                                                       v_PcoToUE,
                                                                                                       v_BearerContextNumber))});
    return v_EpsDefaultBearerId;
  }


1. [bookmark: _Toc89694476]Branches executed
NR Band 41, nea2, Nia2.
1. [bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666][bookmark: _Toc89694477]Execution Log Files
3. [bookmark: _Toc43191966][bookmark: _Toc54877395][bookmark: _Toc89694478]HiSilicon Balong 5000
The HiSilicon Balong 5000 passed this test case on Datang LinkTester 5G Protocol Conformance Test System ECT9610. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test Case Execution log file:
\ TC_11_1_6_NR5GC_N41-FDD3.log
· PIXCIT settings used:
\ Pics(Hisilicon-IRAT).xml
\ Pixit(TC_11_1_6_NR5GC).xml

